Prolog Tutorial 8: Backtracking and Cut

James Power, 1997. 



Prolog differs from imperative languages (like C) in that it concentrates on dealing with facts and rules, rather than sequences of instructions. However, for efficiency, it can sometimes be desirable to add explicit control information to programs - this is the purpose of the cut. 



Analysing Cases

Suppose you were asked to write a Prolog program that would take in someone's exam mark and work out their grade. It might look something like the following: 

  grade(N,first) :- N>=70.

  grade(N,two_1) :- N<70, N>=63.

  grade(N,two_2) :- N<63, N>=55.

  grade(N,third) :- N<55, N>=50.

  grade(N,pass) :- N<50, N>=40.

  grade(N,fail) :- N<40.

While this will work it is a little inefficient. The query grade(75,G) will answer G=first as expected but, once this has been satisfied, Prolog will go back to look for any other solutions. In order to do this it will process all of the other options, failing during the body of the rule in each case. 

If we were implementing this in C, we might try using a "switch" statement as follows: 

  // This code is a little artificial for the purpose of comparison

  int fir(int n) { return n>=70; }

  int fir(int n) { return n<70 && n>=63; }

  //  ... fill in the rest ...

  int fai(int n) { return n<40; }

  switch(n) {

    case(fir(n)): cout << "1st"; break;     

    case(tw1(n)): cout << "2.1"; break;     

    case(tw2(n)): cout << "2.2"; break;     

    case(thi(n)): cout << "3rd"; break;     

    case(pas(n)): cout << "Pass"; break;     

    case(fai(n)): cout << "Fail";

  }

Here we explicitly indicate that after one result has been accepted, we need not look at any of the others at all - this is the purpose of the "break" statement in each branch. 

We can do something similar in Prolog to improve efficiency. Basically, we want to tell Prolog that once it has satisfied one version of the predicate, it need look at no other. Prolog's equivalent of the break statement here is the cut, written "!". 

To eliminate useless backtracking from the above, (and taking advantage of Prolog's order of execution) we can rephrase the program as: 

  grade(N,first) :- N>=70, ! .

  grade(N,two_1) :- N>=63, ! .

  grade(N,two_2) :- N>=55, ! .

  grade(N,third) :- N>=50, ! .

  grade(N,pass) :- N>=40, ! .

  grade(N,fail) :- N<40.

The cut predicate has the effect of telling Prolog not to pass back through this point when it is looking for alternative solutions. Thus, the ``!'' acts as a marker, back beyond which Prolog will not go. When it passes this point all choices that is has made so far are ``set''; i.e. they are treated as though they were the only possible choices. 

Note that the cut always appears where a predicate can appear (never, for example, as arguments to a predicate). It is treated at this level just like any other predicate, and it alwayssucceeds. 

In summary, the effect of the cut is as follows: 

1. Any variables which are bound to values at this point cannot take on other values 

2. No other versions of predicates called before the cut will be considered 

3. No other subsequent versions of the predicate at the head of the current rule will be considered 

4. The cut always succeeds. 

Basically, any more answers to the current query must come from backtracking between the point of the cut and the end of the current rule. 



An Example Of Using The Cut

Save the following knowledge base in a file, and read it into Prolog: 

  holiday(friday,may1).

  weather(friday,fair).

  weather(saturday,fair).

  weather(sunday,fair).

  weekend(saturday).

  weekend(sunday).

  % We go for picnics on good weekends and May 1st

  picnic(Day) :- weather(Day,fair), weekend(Day).

  picnic(Day) :- holiday(Day,may1).

Pose the query: 

  picnic(When).

You should get three answers; make sure you understand where they come from! Note that in order to get this answer, Prolog had to work through exactly one unsuccessful instantiation of When with "friday", before getting it right the second time. 



The First Cut

Now change the definition of picnic to the following: 

  picnic(Day) :- weather(Day,fair), !, weekend(Day).

  picnic(Day) :- holiday(Day,may1).

Now when we pose the query: Picnic(When) Prolog will try to satisfy the sub-goal: 

  weather(When,fair), !, weekend(When).

The first rule for weather is: 

  weather(friday,fair), 

so the new sub-goal becomes: 

  ....., !, weekend(friday).

Prolog passes the cut, and goes on to try to satisfy 

  weekend(friday)

which fails. Previously, it would have backtracked to the last choice point, and gone on with processing 

  weather(saturday,fair)

But now the presence of the cut stops it going back, so it is trapped between the cut and the end of the (failed) predicate. 

The answer now is simply: 

  No.

(Check that this is so...) 



Another Cut

Change the definition of picnic for a second time to get: 

  picnic(Day) :- weather(Day,fair), weekend(Day), !.

  picnic(Day) :- holiday(Day,may1).

With the same query Prolog proceeds as before, until it gets to the sub-goal: 

  ....., weekend(friday), !.

This time we go on to process: 

  weekend(friday)

which fails, and so we go back to the last choice point without meeting the cut. 

Since we also have: 

  weather(saturday,fair).

the new sub-goal becomes: 

  ....., weekend(saturday), !.

This time the whole goal succeeds, and Prolog processes the cut. Since there is a successful answer, Prolog prints out: 

  When = saturday.

However, because it has met the cut, it cannot go back, and so it will not return any extra answers. (Check this...) 



Yet Another Cut

Finally, change the definition of picnic once more, to get: 

  picnic(Day) :- !, weather(Day,fair), weekend(Day).

  picnic(Day) :- holiday(Day,may1).

This time when we ask picnic(When) the first thing we do is to process the cut, and Prolog puts down the "no going back" marker. Any solutions we get from now on have to come from between the "!" and the end of the clause. 

As before 

  weather(friday,fair)

fits, and so we try to satisfy: 

  weekend(friday)

which fails. We backtrack to the last choice point, which was for the goal: 

  weather(Day,fair)

Since we can get back here without passing the cut, we are free to consider the alternatives, and ultimately get: 

  When = saturday.

  When = sunday.

Note that the second attempt to get the answer friday never happens, because getting to the goal for this would involve crossing the cut, which we can't do. 

Thus there are only two solutions in this case. 



Exercises

1. Assume that we have a Prolog program with the following facts: 

2.   p(a).  q(a,1).  r(1,1).  r(3,5). 

3.   p(b).  q(a,2).  r(1,2).  r(3,6). 

4.          q(b,3).  r(2,3).  r(4,7). 

5.          q(b,4).  r(2,4).  r(4,8). 

What are the results of running the following queries? 

1. p(X), q(X,Y), r(Y,Z). 
2. !, p(X), q(X,Y), r(Y,Z). 

3. p(X), !, q(X,Y), r(Y,Z). 

4. p(X), q(X,Y), !, r(Y,Z). 

5. p(X), q(X,Y), r(Y,Z), !. 

6. Consider the following program which is intended to define the third argument to be the maximum of the first two numeric arguments: 

7.   max(X,Y,X) :- X >= Y, !.

8.   max(X,Y,Y).

1. Provide an appropriate query to show that this program is incorrect (try using all constant arguments) 

2. Change the program so that it works correctly 

9. Consider the following program which is supposed to insert its first argument, a number, into its second argument, a sorted list, giving the third argument (also a sorted list): 

10.   insert(X,[H|T],[H|T1]) :- X>H, !, insert(X,T,T1).

11.   insert(X,L,[X|L]).

1. Provide an appropriate query to show that this program is incorrect 

2. Change the program so that it works correctly 

